

## 9/9.3

# Transformez votre Amstrad CPC + DMP 2000 en machine à écrire

---

Sur tous les Amstrad CPC existent des traitements de texte évolués (SEMWORD, TASWORD, POCKET WORDSTAR, TEXTOMAT, et votre traitement de texte WEKA) qui traitent très facilement le courrier.

Il est tout de même fastidieux de lancer un logiciel demandant une à plusieurs minutes de chargement pour une unique lettre de quelques malheureuses lignes, et ne demandant pas de présentation exceptionnelle.

Nous vous proposons un petit logiciel permettant d'écrire une lettre rapidement avec un contrôle au fur et à mesure de l'impression.

Les seules options que nous avons retenues sont :

- l'appui sur la touche <TAB> qui effectue une tabulation de lignes ;
- la touche <DEL> qui permet de revenir en arrière d'un caractère ;
- la touche de soulignement (située au-dessus du zéro) qui fonctionne comme une bascule : lorsque l'on appuie une fois, le texte entré ensuite est automatiquement souligné : si l'on appuie une deuxième fois, le soulignement est supprimé ;
- la touche <RETURN> effectue un passage à la ligne suivante avec retour de la tête d'impression sur la gauche de la feuille de papier.

Afin de transformer notre matériel en machine à écrire, il faut que les caractères frappés au clavier soient automatiquement imprimés. Nous utilisons à cet effet la fonction de l'imprimante appelée IMPRESSION INCREMENTIELLE et sélectionnée par le code.

**ESC i + 1**

c'est-à-dire par l'envoi de la chaîne

**CHR\$(27) ; « i » ; CHR\$(1)**

Ce code permet aux caractères envoyés à l'imprimante d'être imprimés immédiatement. Dès que l'impression est effectuée, le papier avance

automatiquement de quelques lignes, afin de vous permettre de lire le texte inscrit, puis revient en position dès qu'un autre caractère doit être imprimé.

*Remarque :*

L'annulation de cette séquence s'effectue par la commande **ESC i + O**).

L'utilisation, ainsi que le programme lui-même étant particulièrement simple, l'algorithme correspondant sera aussi des plus simples :

- DEBUT
    - Ré-initialiser l'imprimante
      - CO  
Au cas où il lui aurait été envoyé d'autres codes de contrôle
      - FINCO
    - Initialiser l'imprimante en mode incrémentiel
    - Initialiser l'imprimante en qualité courrier
    - TANT QUE la touche <ESC> n'est pas enfoncée deux fois
      - Attendre le caractère au clavier
      - SI c'est un caractère de contrôle
        - CO  
soulignement-retour chariot...
        - FINCO
      - ALORS
      - Envoyer le code correspondant
      - Visualiser à l'écran
      - Retourner acquérir un caractère
    - FINSI
    - Envoyer
  - FINTANTQUE
  - Ejecter la feuille
  - Ré-initialiser l'imprimante
- FIN

Le listing correspondant est le suivant :

```
10 REM *****
20 REM * MACHINE A ECRIRE *
30 REM * SUR AMSTRAD *
40 REM *****
50 '
60 ON BREAK GOSUB 330
70 REM *** REINITIALISATION DE L'IMPRIMA
NTE ***
80 PRINT#8,CHR$(27);"à"
90 '
100 REM *** INITIALISATION IMPRIMANTE EN
MACHINE A ECRIRE ***
110 PRINT#8,CHR$(27);"i";CHR$(1)
120 '
130 REM *** PASSAGE EN QUALITE COURRIER
***
140 PRINT#8,CHR$(27);"x";CHR$(1)
150 '
160 REM *** MACHINE EN FONCTIONNEMENT *
**
170 A$ = INKEY$
180 IF A$ = "" THEN 170
190 IF A$ = CHR$(13) THEN A$ = A$+CHR$(1
0)
200 IF A$ = CHR$(127) THEN A$ = CHR$(8)
210 IF A$ <> CHR$(95) THEN 280
220 REM * SOULIGNEMENT *
230 IF FLAG = 0 THEN PRINT#8,CHR$(27);"-
";CHR$(1);:FLAG = 1:PRINT CHR$(24);:GOTO
170
240 PRINT#8,CHR$(27);"-";CHR$(0);
250 FLAG = 0
260 PRINT CHR$(24);
270 GOTO 170
280 PRINT#8,A$;
290 PRINT A$;
300 GOTO 170
310 '
320 REM *** FIN DOCUMENT ***
330 PRINT#8
340 '
350 REM *** REINITIALISATION IMPRIMANTE
***
360 PRINT#8,CHR$(27);"à"
370 '
380 REM *** SAUT DE PAGE ***
390 PRINT#8,CHR$(12)
400 '
410 STOP
```

- Ligne 110 : transformation du retour chariot en un retour chariot plus passage à la ligne, sinon la tête d'imprimante reviendrait à gauche, et la suite du texte serait imprimé sur la ligne précédente.
- Ligne 200 : traitement de la touche <DEL> en un retour arrière d'un caractère.
- Lignes 220 à 270 : traitement du soulignement avec la création d'une variable FLAG pour savoir si le soulignement est actif ou non. La ligne 260 permet de visualiser le texte à l'écran en vidéo inversée si celui-ci est souligné.
- Lignes 280 et 390 : le caractère est envoyé à l'imprimante et à l'écran.
- Ligne 330 : vidage du tampon imprimante.
- Lignes 360 à FIN : ré-initialisation imprimante et saut de page.